

Traditional Omnibus Mediation Tests for Social Phobia Example

This document considers in more depth the output from Mplus for the overall omnibus test of mediation for the social phobia example in Chapter 11. By an omnibus effect, I mean estimating the effect of the treatment on the outcome through a specific mediator. I report how to obtain the omnibus tests for the FISEM approach as well as interventional indirect effects introduced in Chapter 9. I then describe a method for obtaining omnibus mediation effects using OLS-based LISEM.

For convenience, I repeat the original influence diagram from the main text of Chapter 11 (absent covariates to avoid clutter), followed by the core model equations that include the covariates:




FIGURE 1. Social phobia example

Here are the core equations for the model using sample notation (I use short labels for the variable concepts to save space; I use somewhat different labels later for the *measures* of the concepts. The codes are T = treatment condition, PSS = perceived social skills, NCA = negative cognitive appraisals, ELC = external locus of control, LSP = latent social phobia,

BS = biological sex, PH = parental hypercriticism):

$$\text{NCA2} = a_1 + p_1 T + p_8 \text{PSS2} + b_1 \text{BS1} + b_2 \text{PH1} + b_3 \text{NCA1} + d_1 \quad [1]$$

$$\text{PSS2} = a_2 + p_2 T + b_4 \text{BS1} + b_5 \text{PH1} + b_6 \text{PSS1} + d_2 \quad [2]$$

$$\text{ELC2} = a_3 + p_3 T + p_9 \text{PSS2} + b_7 \text{BS1} + b_8 \text{PH1} + b_9 \text{ELC1} + d_3 \quad [3]$$

$$\text{LSP3} = a_4 + p_7 T + p_4 \text{NCA2} + p_5 \text{PSS2} + p_6 \text{ELC2} + b_{10} \text{BS1} + b_{11} \text{PH1} + b_{12} \text{LSP1} + d_4 \quad [4]$$

$$\text{CR3} = a_5 + L_1 \text{LSP3} + e_1 \quad [5]$$

$$\text{SPAI3} = a_6 + L_2 \text{LSP3} + e_2 \quad [6]$$

$$\text{SPIN3} = a_7 + L_3 \text{LSP3} + e_3 \quad [7]$$

$$\text{CR1} = a_8 + L_4 \text{LSP1} + e_4 \quad [8]$$

$$\text{SPAI1} = a_9 + L_5 \text{LSP1} + e_5 \quad [9]$$

$$\text{SPIN1} = a_{10} + L_6 \text{LSP1} + e_6 \quad [10]$$

In the above equations, I use p notation for the path coefficients, b notation for coefficients associated with covariates, and L notation for unstandardized factor loadings.

Finally, here is the Mplus syntax I used for the FISEM analysis, per Chapter 11:

Table 1: Mplus Syntax for Social Phobia Example

```

1. TITLE: EXAMPLE CHAPTER 11 ;
2. DATA: FILE IS c:\mplus\ret\chap11M.txt ;
3. VARIABLE:
4. NAMES ARE ID CR1 SPAI1 SPIN1 CR3 SPAI3 SPIN3
5. NEGAPP2 PSKILLS2 EXTERN2 NEGAPP1 PSKILLS1 EXTERN1
6. HYPER SEX TREAT ;
7. USEVARIABLES ARE CR1 SPAI1 SPIN1 CR3 SPAI3 SPIN3
8. NEGAPP2 PSKILLS2 EXTERN2 NEGAPP1 PSKILLS1 EXTERN1
9. HYPER SEX TREAT ;
10. MISSING ARE ALL (-9999) ;
11. ANALYSIS:
12. ESTIMATOR = MLR ; !Robust maximum likelihood
13. MODEL:
14. !Specify latent variables
15. LSP1 BY CR1 SPAI1 SPIN1 ;
16. LSP3 BY CR3 SPAI3 SPIN3 ;
17. [CR1@0] ; [CR3@0] ; [LSP1] (mean1) ; [LSP3] (int1) ;
18. !Specify equations
19. LSP3 ON LSP1 NEGAPP2 PSKILLS2 EXTERN2 TREAT SEX (b10 p4-p7 b11) ;
20. LSP3 ON HYPER (b12) ;
21. NEGAPP2 ON TREAT HYPER SEX NEGAPP1 PSKILLS2 (p1 b1-b3 p8) ;

```

```

22. PSKILLS2 ON TREAT HYPER SEX PSKILLS1 (p2 b4-b6) ;
23. EXTERN2 ON TREAT HYPER SEX EXTERN1 PSKILLS2 (p3 b7-b9 p9) ;
24. !Specify correlations of latent variable with exogenous variables
25. LSP1 WITH NEGAPP1 PSKILLS1 EXTERN1 TREAT SEX HYPER ;
26. MODEL INDIRECT:
27. LSP3 IND TREAT ;
28. LSP3 IND PSKILLS2 ;
29. NEGAPP2 IND TREAT ;
30. EXTERN2 IND TREAT ;
31. OUTPUT:
32. SAMP STANDARDIZED(STDYX) MOD(ALL 4) RESIDUAL CINTERVAL TECH4 ;

```

FISEM ANALYSIS OF OMNIBUS MEDIATION

The syntax from Table 1 that generates results for the analysis of mediational chains is on lines 26 to 28. I focus on Lines 26 and 27. Popular estimation algorithms for evaluating omnibus mediation effects include MLR and percentile bootstrapping, with the latter probably being the better of the two for the case of all continuous variables. The bootstrap approach does not permit modification indices in Mplus, so I usually first evaluate the fit of my model using MLR to ensure I have a good fitting model based on inspection of both global and localized fit indices, including modification indices. Once I have verified reasonable model fit, I re-run the program using bootstrapping, with the syntax adjustments described in Chapter 11 to accommodate a bootstrap solution. I often find there is little difference between the MLR and bootstrapped results for parameter estimates and their standard errors, in which case I can report either of them so as to maintain consistency throughout my write-up. However, I am always careful to check for differences between the MLR and bootstrap analyses to see if non-trivial differences emerge. If they do, I use the bootstrap method, unless the sample size is too small to accommodate bootstrapping (see Chapter X). In the social phobia example, because the results were so similar for the MLR and bootstrap analyses, I use the MLR output, which is what I used in Chapter 11.

For the unstandardized omnibus mediation tests, the results are in the section called TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS. Here is the first part of the output:

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Effects from TREAT to LSP3				
Total	-1.758	0.104	-16.855	0.000
Total indirect	-1.270	0.121	-10.463	0.000

The first row estimates the total effect of the variable to the left of the line above it, `TREAT`, on the variable to the right on that line, `LSP3`. The effect of the treatment relative to the control group was to lower social phobia by -1.758 ± 0.21 ($z = 16.86$, $p < 0.05$). Below this line is the amount of this effect that is due to all of the mediators considered simultaneously, in this case, perceived social skills, negative cognitive appraisals and external locus of control. The three mediators multivariately lowered social phobia by -1.270 ± 0.24 ($z = 10.46$, $p < 0.05$). The difference between the values of -1.758 and -1.270 is the amount of the overall effect that is due to the direct effect of the program on social phobia independent of the mediators, $1.758 - 1.270 = 0.488$.

Next, `Mplus` provides results for different mediational chains in the model:

Specific indirect 1				
LSP3				
NEGAPP2				
TREAT	-0.234	0.062	-3.766	0.000
Specific indirect 2				
LSP3				
PSKILLS2				
TREAT	-0.829	0.123	-6.735	0.000
Specific indirect 3				
LSP3				
EXTERN2				
TREAT	0.000	0.002	-0.017	0.986
Specific indirect 4				
LSP3				
NEGAPP2				
PSKILLS2				
TREAT	-0.208	0.053	-3.913	0.000
Specific indirect 5				
LSP3				
EXTERN2				
PSKILLS2				
TREAT	0.001	0.036	0.017	0.986

There are a total of 5 mediational chains linking `TREAT` to `LSP3`. Consider `Specific indirect 1`. The variable in the last row underneath this heading is assumed to influence the variable in the next to last row which, in turn, influences the variable in the top row. So, this chain refers to $TREAT \rightarrow NEGAPP2 \rightarrow LSP3$. We work our way from top to bottom of the output to isolate the mediational chain. The omnibus coefficient *for this particular chain* is -0.234 ± 0.12 ($z = 3.76$, $p < 0.05$). Because `TREAT` is a dummy variable with dummy

coding, the coefficient is the mean social phobia difference between the intervention condition minus the control condition through the mediational chain focused on negative cognitive appraisals. From Figure 11.2 in Chapter 11, it is p_1 times p_4 . Because the coefficient is negative, this means a larger number was subtracted from a smaller number, so the intervention reduced social phobia by -0.234 units through the mediator of negative cognitive appraisals. Mplus reports in the remaining statement corresponding output for all the other omnibus indirect effects through which `TREAT` impacts `LSP3`.

There is an important qualification, however. The `Specific indirect effect 1` in the output does *not* reflect the full omnibus mediational effect between `TREAT` and `LSP3` through `NEGAPP2`. This is because it ignores the chain `TREAT` → `PSKILLS2` → `NEGAPP2` → `LSP3`, which is listed in `Specific indirect effect 4`. The latter has a value of -0.208 ± 0.11 ($z = 3.91$, $p < 0.05$). To obtain the full omnibus effect of `TREAT` on `LSP3` through `NEGAPP2`, I need to sum *all* of the reported indirect effect chains that have `NEGAPP2` in them, in this case `indirect effect 1` and `indirect effect 4`. This yields $-0.234 + -0.208 = -0.442$. That is, using the notation in Figure 11.2, I calculate

$$p_1 p_4 + p_2 p_8 p_4$$

If I want to obtain a significance test and confidence interval of this combined omnibus mediation effect through `NEGAPP2`, I can use the `MODEL CONSTRAINT` feature of Mplus. I add the following syntax just before the output line:

```
MODEL CONSTRAINT:
NEW(OME1);
OME1 = p1*p4 + p2*p8*p4 ;
```

The first line tells Mplus I want to calculate combinations of paths. The second line tells Mplus I am going to calculate a new parameter that I will label `OME1` (an acronym for omnibus mediation effect 1). You can use any label you want but it must not exceed 8 characters and it must follow Mplus conventions. The third line defines the new parameter using the labels assigned to the relevant paths in the Mplus syntax (see Table 11.1 in Chapter 11). The output for the above appears in the section `MODEL RESULTS` under the subsection `New/Additional Parameters`. Here are the results:

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
OME1	-0.442	0.106	-4.171	0.000

The effect of TREAT on LSP3 through the mediational chain of NEGAPP2 is to lower social phobia by -0.442 ± 0.21 ($z = 4.17, p < 0.05$).

Mplus offers syntax that allows one to do the above in a simpler way instead of using the MODEL CONSTRAINT command. To the main syntax in Table 1, I can add under MODEL INDIRECT on a new line (say after Line 30) the command

```
LSP3 VIA NEGAPP2 TREAT ;
```

This requests that Mplus calculate all of the indirect effects between TREAT and LSP3 that go through NEGAPP2 as well as the sum of them. Here is the output:

```
Effects from TREAT to LSP3 via NEGAPP2
```

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Sum of indirect	-0.442	0.106	-4.171	0.000
Specific indirect 1				
LSP3				
NEGAPP2				
TREAT	-0.234	0.062	-3.766	0.000
Specific indirect 2				
LSP3				
NEGAPP2				
PSKILLS2				
TREAT	-0.208	0.053	-3.913	0.000

Mplus routinely prints out standardized coefficients and confidence intervals for all of the above, except it does not provide standardized coefficients for the new parameters defined in the MODEL CONSTRAINT command. Also, when using the MODEL CONSTRAINT command you must remove the request for modification indices on the output line. Given this, I execute the MODEL CONSTRAINT commands as a supplementary analysis after I have already established good model fit using syntax without the MODEL CONSTRAINT command.

The same principles I outline above for NEGAPP2 can be used to obtain and test for omnibus mediation effects for the other mediators. Here are the MODEL CONSTRAINT command that estimates all three omnibus mediation effects in the social phobia example:

```
MODEL CONSTRAINT:
NEW (ONEGAPP2 OPSKILLS2 PEXTERN2) ;
ONEGAPP2 = p1*p4 + p2*p8*p4 ;
OPSKILLS2 = p2*p5 + p2*p8*p4 + p2*p9*p6 ;
```

```
OEXTERN2 = p3*p6 + p2*p9*p6 ;
```

I use the letter O at the beginning of each label to stand for “omnibus,” but you can use any 8 character label you want. Instead of ME in the label, I use the mediator name to help me identify it on the output. Alternatively, I can use three VIA commands, as follows:

```
LSP3 VIA NEGAPP2 TREAT ;
LSP3 VIA PSKILLS2 TREAT ;
LSP3 VIA EXTERN2 TREAT ;
```

These steps are necessary because the overall model has causal relationships among the mediators. If this was not the case, I would not need any of the `MODEL CONSTRAINT` commands or the `VIA` commands because each of the mediators would appear in only one chain. Everything you need would be self-contained in the Mplus output. Such an example is presented in Chapter 12. Note that the omnibus tests for a given mediational chain, in my opinion, do not add much in the way of new information about the three key questions of an RET addressed in Chapter 11, namely (1) does the program have an overall effect on the outcome, (2) does the program affect each of the targeted mediators, and (3) is each of the target mediators relevant to the outcome. If you find the omnibus analyses helpful beyond the joint significance test, then by all means pursue them. Parenthetically, Tofighi and Kelley (2020) evaluated the relative merits of different approaches to calculating confidence intervals for mediation effects that involve sequential mediation with multiple mediators, such as where $T \rightarrow M1 \rightarrow M2 \rightarrow Y$. Such mediation is present in the current example. They found that for $N = 50$, Bayesian estimation tended to perform best but for $N > 100$, the percentile bootstrap performed as well as the Bayesian methods.

INTERVENTIONAL INDIRECT EFFECTS

As discussed in Chapters 9 and 10, Loh et al. (2022) describe a form of mediation analysis called **interventional indirect effects** (see also Didelez et al., 2006; Hayes, 2018; VanderWeele et al., 2014; Vansteelandt & Daniel 2017). The approach focuses on multiple mediator models and seeks to estimate an omnibus mediation effect for a given mediator but for the case where the causal structure among the multiple mediators is unknown. Because the social phobia example in Chapter 11 specifies causal relationships among the mediators, I can use it to provide perspectives on interventional indirect effects. For good introductory treatments of interventional indirect effects, see Nguyen, Schmid and Stuart (2021); Loh and Dongning (2022), and Loh et al., (2022).

The concept of an interventional indirect effect is strongly tied to the causal mediation framework described in Chapter 9; it draws heavily on potential outcome conceptualizations

of causality. Interventional indirect effect analysis evolves from the non-trivial challenges of testing multiple mediator models in the causal mediation/potential outcomes framework. Interventional indirect effects are defined differently from traditional indirect effects in mediation analysis, which can be confusing to some. I introduce the approach for the simplest case of all continuous mediators and a continuous outcome for an RET.

The calculation of interventional indirect effects makes three core assumptions. The first assumption is that the effect of the treatment condition on the outcome is unconfounded by unobserved/unmeasured covariates, i.e., that the observed covariates are sufficient to adjust for confounding. This assumption typically is met in randomized trials. The second assumption is that the effect of the mediators on the outcome are unconfounded by unobserved or unmeasured covariates. The third assumption is that the effect of the treatment condition on the mediators are unconfounded by unobserved/unmeasured covariates. These assumptions also are typically made in more traditional mediation analyses.

In the social phobia example, consider the mediator negative cognitive appraisals, `NEGAPP2`. Calculation of the omnibus mediation effect for `TREAT` to `LSP3` through `NEGAPP2` requires that we take into account two pathways when analyzing the first link in the mediational chain (the effect of `TREAT` on `NEGAPP2`), namely the direct effect `TREAT` \rightarrow `NEGAPP2` and the indirect effect of the treatment on negative appraisals through perceived social skills, `TREAT` \rightarrow `PSKILLS2` \rightarrow `NEGAPP2`. In traditional mediation analysis, my estimate of the overall effect of the treatment condition on the target mediator will be unbiased as long as these causal relationships among the mediators are correctly specified relative to the true underlying population dynamics. However, suppose they are not. Suppose that instead of the links `TREAT` \rightarrow `NEGAPP2` and `TREAT` \rightarrow `PSKILLS2` \rightarrow `NEGAPP2` being true, the causal structure among the mediators is `TREAT` \rightarrow `NEGAPP2` and `TREAT` \rightarrow `NEGAPP2` \rightarrow `PSKILLS2`. The approach of Loh et al. still can estimate the overall effect of `TREAT` on the negative appraisal mediator (or some other mediator in the model) under such misspecification because, as I will show shortly, it does not require knowledge of the true causal relationships among the mediators when deriving such estimates. As long as you do not care about what the mediational causal structure is and all you want to know is the overall effect of the treatment on a given mediator for purposes of defining the first link in the mediational chain, the Loh et al. approach can be informative.

For the social phobia example, the interventional indirect effect approach essentially avoids specifying causal dependence among mediators by modeling each mediator as only depending on the treatment condition, `TREAT`, and the covariates designed to control for confounds. This means that in Figure 1, p_1 is retained in the model as is, but paths p_8 and p_9 are dropped from the model. Doing so will inflate the value of p_1 by the value of $p_2 * p_8$

because the latter will be absorbed into p_1 . Similarly, the value of p_3 will be inflated by the value of $p_2 * p_9$ when estimating the effect of the treatment on external locus of control because $p_2 * p_9$ is absorbed into p_3 . I will refer to these reparameterized p_1 and p_3 parameters as p_{1intde} and p_{3intde} to signify they are part of the interventional indirect effect rather than the more classic indirect effect specification. I also will specify p_2 as p_{2intde} to indicate that it too is part of the interventional indirect effect analysis so as to maintain consistency in model notation. This reparameterization of the effects of the treatment condition on each mediator allows us to estimate $TREAT \rightarrow MEDIATOR$ effects without knowing the causal structure among the mediators.

The interventional indirect effect approach becomes more controversial when mapping the second link in the mediational chain, namely the effects of a given mediator on the outcome per paths p_4 through p_6 in Figure 1. As noted earlier in this chapter, to estimate the effect of $PSKILLS2$ on $LSP3$, we need to take three pathways into account (1) the direct effect of $PSKILLS2$ on $LSP3$, namely p_5 , (2) the indirect effect of $PSKILLS2$ on $LSP3$ through negative appraisals (i.e., $p_8 * p_4$), and (3) the indirect effect of $PSKILLS2$ on $LSP3$ through external locus of control ($p_9 * p_6$). The interventional indirect effect, however, uses only the first term (the direct effect of $PSKILLS2$ on $LSP3$) for the second link and ignores the two indirect effects. In the social phobia example, this constitutes a misspecified model because I generated the data for it as coming from a population model that takes the form of Figure 1. Loh and Dongning (2022) argue that using only the direct effect of a mediator on the outcome is not necessarily unreasonable if the researcher truly has no clue or theoretical leads as to the operative causal relationships among the mediators in the population. They further argue that, taken together with the estimation of the $TREAT \rightarrow PSKILLS2$ logic described earlier, the multiplication of the two paths (p_{2intde} and p_5) still captures the essence of the meaning of mediation, namely it reflects the shift in the mediator caused by the treatment weighted by the shift in outcome directly caused by the mediator. Essentially, an interventional indirect effect is defined differently than traditional indirect effects. Specifically, an interventional indirect effect via a mediator is the combined effect along all underlying causal pathways leading from the treatment condition to the mediator of interest weighted by the causal pathway from the mediator directly to the outcome. By contrast, the traditional indirect effect via a mediator is the combined effect along all underlying causal pathways leading from the treatment condition to the mediator of interest weighted by all underlying causal pathways leading from the mediator to the outcome. Interventional indirect effects are distinct from traditional indirect effects and require unique interpretations accordingly.

Advocates of the use of interventional indirect effects articulate different scenarios where the approach has utility relative to the more traditional approach to indirect effects. For

example, if a researcher is truly clueless about the causal relationships among mediators, then a researcher might be better advised to use the interventional indirect effect approach rather than an arbitrary model with difficult-to-justify causal relationships among the mediators that might be misspecified in their own right. In my own work, I am rarely completely clueless about reasonable causal structures among mediators given common sense and past research in my substantive domains. I personally prefer to examine estimates under different substantively reasonable model scenarios and see how my conclusions might change or not change as a function of model specification rather than resort to interventional indirect effects. However, I also believe that sometimes the interventional indirect effect approach is viable. I refer you to Nguyen et al. (2021); Loh and Dongning (2022), Loh et al., (2022) and the initial work on interventional indirect effects (e.g., VanderWeele et al., 2014; Vansteelandt & Daniel 2017) for further consideration of approach viability, applications, and additional assumptions that the approach makes.

Table 2 presents the Mplus syntax I would use to estimate interventional indirect effects for the social phobia example. As recommended by interventional direct effect advocates, I use bootstrapping when estimating effects (see Lines 12 and 32). Note on Lines 21 to 23, I only include the direct effect of the treatment on each mediator and ignore any indirect effects. Lines 26 to 30 specify the interventional indirect effects. The estimates and their significance tests appear on the output under `ADDITIONAL PARAMETERS`.

Table 2: Mplus Syntax for Interventional Indirect Effects for Social Phobia Example

```

1. TITLE: EXAMPLE CHAPTER 11 ;
2. DATA: FILE IS c:\mplus\ret\chap11M.txt ;
3. VARIABLE:
4. NAMES ARE ID CR1 SPAI1 SPIN1 CR3 SPAI3 SPIN3
5. NEGAPP2 PSKILLS2 EXTERN2 NEGAPP1 PSKILLS1 EXTERN1
6. HYPER SEX TREAT ;
7. USEVARIABLES ARE CR1 SPAI1 SPIN1 CR3 SPAI3 SPIN3
8. NEGAPP2 PSKILLS2 EXTERN2 NEGAPP1 PSKILLS1 EXTERN1
9. HYPER SEX TREAT ;
10. MISSING ARE ALL (-9999) ;
11. ANALYSIS:
12. ESTIMATOR = ML ; BOOTSTRAP=5000 ; !Use bootstrapping
13. MODEL:
14. !Specify latent variables
15.     LSP1 BY CR1 SPAI1 SPIN1 ;
16.     LSP3 BY CR3 SPAI3 SPIN3 ;
17. [CR1@0] ; [CR3@0] ; [LSP1] (mean1) ; [LSP3] (int1) ;
18. !Specify equations
19. LSP3 ON LSP1 NEGAPP2 PSKILLS2 EXTERN2 TREAT SEX (b10 p4-p7 b11) ;
20. LSP3 ON HYPER (b12) ;
21. NEGAPP2 ON TREAT HYPER SEX NEGAPP1 (p1 b1-b3) ;

```

```

22. PSKILLS2 ON TREAT HYPER SEX PSKILLS1 (p2 b4-b6) ;
23. EXTERN2 ON TREAT HYPER SEX EXTERN1 (p3 b7-b9) ;
24. !Specify correlations of latent variable with exogenous variables
25. LSP1 WITH NEGAPP1 PSKILLS1 EXTERN1 TREAT SEX HYPER ;
26. MODEL CONSTRAINT:
27. NEW (INTIE1 INTIE2 INTIE3) ;
28. INTIE1 = p1*p4 ;
29. INTIE2 = p2*p5 ;
30. INTIE3 = p3*p6 ;
31. OUTPUT:
32. SAMP STANDARDIZED(STDYX) RESIDUAL CINTERVAL(BOOT) TECH4 ;

```

LISEM ANALYSIS OF OMNIBUS MEDIATION

For the OLS-based LISEM analyses, the simplest way to obtain estimates of the omnibus mediation effect is to use the *Monte Carlo CI* program on my webpage. There is a video that accompanies the program that will walk you through what you need to do. You will use the same formulae that you did in the above `MODEL CONSTRAINT` command, namely $p1*p4 + p2*p8*p4$ for negative appraisals, $p2*p5 + p2*p8*p4 + p2*p9*p6$ for perceived social skills, and $p3*p6 + p2*p9*p6$ for locus of control.

REFERENCES

- Didelez, V., Dawid, A. P., & Geneletti, S. (2006). Direct and indirect effects of sequential treatments. In Proceedings of the 22nd conference on uncertainty in artificial intelligence (pp. 138–146). Arlington, VA, USA: AUAI Press
- Hayes, A. (2018). *Introduction to mediation, moderation, and conditional process analysis: A regression-based approach*. Second edition. Guilford.
- Loh, W. & Dongning, R. (2022). Improving causal inference of mediation analysis with multiple mediators using interventional indirect effects. *Social and Personality Compass*, 16, e12708.
- Loh, W., Moerkerke, B., Loeys, T. & Vansteelandt, S. (2022). Disentangling indirect effects through multiple mediators without assuming any causal structure among the mediators. *Psychological Methods*, 27, 982-999.
- Nguyen, T., Schmid, I. & Stuart, E. (2021). Clarifying causal mediation analysis for the applied researcher: Defining effects based on what we want to learn. *Psychological Methods*, 26, 255–271.

Tofighi, D. & Kelley, K. (2020). Indirect effects in sequential mediation models: Evaluating methods for hypothesis testing and confidence interval formation. *Multivariate Behavioral Research*, 55, 188–210.

VanderWeele, T. J., Vansteelandt, S., & Robins, J. M. (2014). Effect decomposition in the presence of an exposure-induced mediator-outcome confounder. *Epidemiology*, 25, 300.

Vansteelandt, S., & Daniel, R. M. (2017). Interventional effects for mediation analysis with multiple mediators. *Epidemiology*, 28, 258-265.